

Maple Cheat Sheet – www.zwiener.org

Analysis

Funktion definieren:

```
f := x -> x^2 + 2*x + 1;
```

$$f := x \rightarrow x^2 + 2x + 1$$

Funktion mehrerer Veränderlichen:

```
f := (x,y) -> 4*y-x^2-y^2;
```

$$f := (x,y) \rightarrow 4y - x^2 - y^2$$

Faktorisieren:

```
factor(f(x));
```

$$(x+1)^2$$

Ausmultiplizieren:

```
expand(f(x));
```

$$x^2 + 2x + 1$$

Vereinfachen:

```
simplify(3*x + 5*x - 10 + 20 - x);
```


$$7x + 10$$

Eigenschaften von Variablen festlegen / Definitionsbereich bestimmen:

```
f := b*Pi/sqrt(1/a^2);  
f;  
assume(a > 0); assume(b > 0);  
simplify(f);
```

$$f := \frac{b\pi}{\sqrt{\frac{1}{a^2}}}$$

$b \sim \pi a$



Variablen als ganze Zahlen:

```
assume(k, integer);
```


e-Funktion, Unendlich und Kreiszahl Pi:

```
exp(1); infinity; Pi;
```

$e \quad \infty \quad \pi$

Graph zeichnen (discont = Sprungstellen berücksichtigen, labels = Achsen beschriften):

```
plot(sin(x), x=0..2*Pi, y=-1..1, labels=[x,y],  
color=[blue], discont=true);
```



Ableitung (als Ausdruck):

```
f := x -> x^2 + 2*x - 10;  
diff(f(x), x);
```

$f := x \rightarrow x^2 + 2x - 10$
 $2x + 2$

Ableitung (als Funktion):

```
g := D(f);
```

$g := x \rightarrow 2x + 2$

Grenzwerte berechnen:

```
limit( sin(x)/x, x=0 ); # left, right als 3. Parameter
```

Unbestimmtes Integral:

```
int(f(x), x);
```

$\frac{1}{3}x^3 + x^2 - 10x$

Bestimmtes Integral:

```
int(f(x), x=0..5);
```

$\frac{50}{3}$

Reihen:

```
sum( a(k), k=0..infinity );
```

Bsp. Geometrische Reihe:

```
assume(q < 1, q > -1);  
about(q);  
sum(q^k, k=0..infinity);
```

$$-\frac{1}{q-1}$$

Potenzreihen (Entwicklungspunkt an Stelle 0 mit 5 Reihenglieder):

```
series(exp(x), x=0,5);
```

$$1+x+\frac{1}{2}x^2+\frac{1}{6}x^3+\frac{1}{24}x^4+O(x^5)$$

Potenzreihe als Funktion:

```
p(x) := convert(%, polynom);
```

$$p(x) := 1+x+\frac{1}{2}x^2+\frac{1}{6}x^3+\frac{1}{24}x^4$$

Quadratische Gleichungen lösen:


```
solve(144*x^2-15*x-10 = 0, x);
```

$$\frac{5}{96} + \frac{\sqrt{665}}{96}, \frac{5}{96} - \frac{\sqrt{665}}{96}$$

Fourier-Reihen:

```
f := x -> signum(sin(x));  
a := k -> 1/Pi * int(f(x)*cos(k*x), x=0..2*Pi);  
assume(k, integer);  
a(k); # immer 0  
b := k -> 1/Pi * int(f(x)*sin(k*x), x=0..2*Pi);  
  
# Zeichnen:  
a(0)/2 + sum(a(k)*cos(k*x) + b(k)*sin(k*x), k=1..5);  
plot([f(x), %], x=-Pi..3*Pi);
```

$$b(k) = \frac{2(1+(-1)^{(1+k)})}{\pi k}$$



Fourier-Transformation:


```
assume(T > 0);
f := t -> piecewise(t < 0, 0, t<=T, 1, 0);
int(f(t)*exp(-I*omega*t), t=-infinity..infinity);
simplify(%);
evalc(%); # Als Real- und Imaginärteil zeigen
```

$$\frac{-I(e^{(\omega T - I)} - 1)e^{-IT - \omega}}{\omega} \quad \text{nach evalc:}$$

$$\frac{\sin(\omega T)}{\omega} + \frac{(-1 + \cos(\omega T))I}{\omega}$$

3D-Plot:

```
f := (x, y) -> x^2 + y^2;
plot3d(f(x, y), x=-2..2, y=-2..2, labels=[x,y,z],
grid=[50,50]);
```



Partielle Ableitungen:

```
f := (x, y) -> x^2 + y^2;  
diff(f(x,y), x); # Nach x Ableiten  
D[1](f); # Funktion erstellen, nach x
```

$$f := (x, y) \rightarrow x^2 + y^2$$
$$2x$$
$$(x, y) \rightarrow 2x$$

Partielles Ableiten an Sprungstelle:

```
limit( (f(0 + dx, 0) - f(0, 0)) / dx, dx = 0);
```

Gradient (Nabla) berechnen:

```
with(linalg);  
grad(f(x,y,z), [x, y, z]);  
  
#Beispiel:  
f := (x, y, z) -> sqrt(x)*exp(y*z)*ln(x*y/z);  
grad(f(x, y, z), [x, y, z]);
```

$$f := (x, y, z) \rightarrow \sqrt{x} e^{(yz)} \ln\left(\frac{xy}{z}\right)$$
$$\left[\frac{1}{2} \frac{e^{(yz)} \ln\left(\frac{xy}{z}\right)}{\sqrt{x}} + \frac{e^{(yz)}}{\sqrt{x}}, \sqrt{x} z e^{(yz)} \ln\left(\frac{xy}{z}\right) + \frac{\sqrt{x} e^{(yz)}}{y}, \sqrt{x} y e^{(yz)} \ln\left(\frac{xy}{z}\right) - \frac{\sqrt{x} e^{(yz)}}{z} \right]$$

Ableitung in Richtung:

```
with(linalg);  
e := <3/5, 4/5>; f := (x,y) -> 2*x^2 - y;  
dotprod(grad(f(x,y), [x, y]), e);
```

$$e = \begin{bmatrix} \frac{3}{5} \\ \frac{4}{5} \end{bmatrix}$$

$$f := (x, y) \rightarrow 2x^2 - y$$
$$-\frac{4}{5} + \frac{12x}{5}$$

Mehrfach partiell ableiten:

```
f := (x,y,z) -> z*x*y^2+y*exp(x*y/z);  
diff(f(x,y,z), x$2); # 2 mal nach x  
diff(f(x,y,z), x,y); # xy / yx (siehe Satz Schwarz)  
D[2,3](f);  
D[3,2](f);  
D[3$2](f); # 2 mal nach z
```

Hesse Matrix:

```
with(linalg);  
f := (x,y,z) -> x^2+y*z;  
hessian(f(x,y,z), [x,y,z]);
```

$$\begin{bmatrix} 2 & 0 & 0 \\ 0 & 0 & 1 \\ 0 & 1 & 0 \end{bmatrix}$$

Nicht lineare Gleichungssysteme:

```
f := (x,y) -> x*y - x^2*y - x*y^2;  
fx := D[1](f);  
fy := D[2](f);  
solve({fx(x,y)=0, fy(x,y)=0}, {x,y});
```

$$\{x=0,y=0\}, \{x=1,y=0\}, \{x=0,y=1\}, \{y=\frac{1}{3}, x=\frac{1}{3}\}$$

Evtl. auftretende RootOf(x) (Nullstellen von x) Funktionen als Werte anzeigen lassen:

```
allvalues(RootOf(-3+_Z^2,label = _L11));
```

$$\sqrt{3}, -\sqrt{3}$$

Beispiel Kurvendiskussion:

```
f := (x,y) -> x*y - x^2*y - x*y^2:
with(linalg);
hessian(f(x,y), [x,y]);
# Werte in Hesse Matrix setzen (siehe NLG):
M := subs(x=1/3, y=1/3, %);
det(M); # Determinante oder gleich folgendes:
# Definitheit bestimmen
definite(M, negative_def); #oder positive_def

#Zeichnen:
plot3d(f(x,y),x=-1.5..1.5, y=-1.5..1.5,
labels=[x,y,z], axes=boxed);
```

Binomialkoeffizienten und Fakultät:

```
5!; # 5 Fakultät = 120

binomial(a, k); # a über k

# Bsp.
binomial(1/2, 3);

# entspricht:
(1/2*(1/2 - 1)*(1/2 - 2))/3!;
```

$$\frac{1}{16}$$

Gleichungen umformen nach einer Unbekannten:

```
solve(1/r = 1/x + 1/y, r);
# Diese Gleichung nach r auflösen
```

$$\frac{xy}{y+x}$$